Xnumbers.dll tutorial

General description

The current, industry-wide standard precision for personal computers is 32 bits, which is equivalent to approximately eight decimals. Double precision, standard in Excel and the default in VBA, has a precision of 64 bits, or about 15 decimals. For most problems this is completely satisfactory, but there are some mathematical operations, such as the multiplication or inversion of large matrices, that require better precision and/or special approaches to keep truncation errors from accumulating in the answer. For such problems, as may be encountered in least squares fitting to multi-term polynomials, or in multivariate applications, extended precision may be needed. It may also be useful as a diagnostic tool in program development and testing, to pinpoint precisely where numerical accuracy can be lost. And as scientists increasingly learn how to use Excel for non-trivial data analysis tasks, the need for extended precision will only increase.    

 Xnumbers.dll is an ActiveX dynamic link library, compatible with Windows 2000, NT, and XP, which allows Excel macros to operate in extended precision, of up to 200 decimals. It was created by Leonardo Volpi and his students, and is freely downloadable from http://digilander.libero.it/foxes. It handles numbers internally as character strings, divides them into packets of up to six decimals, and then manipulates these packets to perform elementary mathematical operations such as addition, subtraction, multiplication, and division. More complicated functions then use these elementary functions as building blocks. In the end, the strings are converted back to numbers. The method used is similar to that in the Fortran77 program MPFUN and in its recent successor, ARPREC, which allows arbitrary precision in Fortran90 and C++. For a description of MPFUN see D. H. Bailey, ACM Trans. Math. Software 19 (1993) 288-319; for ARPREC, see http://crd.lbl.gov/~dhbailey/mpdist/. 

One of the beauties of Xnumbers.dll is that it allows the user to modify existing macros while keeping its user interface untouched. In other words, all high-precision operations are performed in the background, and the software user need not see a difference in either input or output. (High-precision results can of course be displayed in full when so desired, as in the code example of Test 2b shown below). Xnumbers is free, it is powerful, and it is fun, but it does slow down the computation considerably. Extended-precision macros are therefore not recommended for routine use, and they are best labeled as such, e.g., with the prefix x, as we have done with the macros in the xMacroBundle. .

Installation

You will need to have an unzip routine handy, because the download is a zip file, and an Adobe Acrobat reader, because the documentation comes as a pdf (portable document format) file. Both of these can be freely downloaded from the web, and should already be on your desktop as standard tools. 

Here is how you can install the dynamic link library. (A specific location is selected here for your convenience, but you can of course place it somewhere else, in which case you should make the corresponding changes in what follows.) Go to the website http://
digilander.libero.it/foxes, select Downloads, and download the xnumberdll.zip to your desktop. Unzip the file, and store its three components, readme, xnumbers.dll, and Xnumbers_DLL_Ref, on the desktop. The readme file contains nothing important: read it and either discard or store somewhere. The reference file, Xnumbers_DLL_Ref, is very worthwhile and has many useful details. You may want to print it out, but please do that later, because it is a 115-page tome. For now we will focus on xnumber.dll. 

On your desktop, click on My Computer ( Local Disk (C:) ( Program Files ( Microsoft Office ( Office (Office11 in Excel2003) ( Library. Then press File ( New ( Folder, and change the name of the so generated folder from New Folder to Xnumbers, then open that file (which, as a new file, should be empty). If necessary, reduce the space taken up by this folder and/or move it, so that you can find xnumbers.dll on the desktop. Click on xnumbers.dll, copy it into the open folder, and close the folder. Now, xnumbers.dll is stored safely, and presumably in good company, next to Solver.

In order to make sure that Excel can find it, open Excel, go to the Visual Basic Editor with Tools ( Macro ( Visual Basic Editor or, faster, with Alt(F11 (on the Mac: Opt(F11), and then, in the VBEditor menu bar, select Tools ( References. This will display the dialog box for the VBA Project. Click on Browse, select Files of type: Type Libraries (*.olb; *.tlb; *.dll), in the Add Reference dialog box go to My Computer ( Local Disk (C:) ( Program Files ( Microsoft Office ( Office ( Library ( Xnumbers, open the file, and type Xnumbers.dll in the Filre name window if it is empty. Click Open it, which will bring you back to the References – VBA Project dialog box. Now use the Priority Up button to ease xnumbers up the list so that it is contiguous with the other tick-marked items. Click OK. From now on, Excel will know where to find xnumbers.dll.  
Using Xnumbers in Excel VBA

Before the extended-precision instructions and functions (here abbreviated as Xfunctions, Xinstructions, etc.) can be used in any particular function, macro, or subroutine, the class of Xnumbers must first be invoked (in computerese: “instantated”) with the line 


Dim MP As New Xnumbers

and, at the end of the code, deactivated with


Set MP = Nothing

Thereafter, individual Xfunctions can be called between those two statements with the prefix MP. followed by the specific Xinstruction, as in 

Sub Test1a()

Dim a, b, c
           

Dim MP As New Xnumbers

a = "1234567890.0987654321"

b = MP.xInv(a)
c = MP.xExp(a, 40)
Debug.Print "a = ", a

Debug.Print "1 / a = ", b

Debug.Print "e ^ a = ", c

Set MP = Nothing

End Sub

which will display the values of a, b = 1/a, and c = ea in the Immediate Window. (You can open the Immediate Window in the VBEditor, with Ctrl(G.) For 1/a in the above example you will obtain answers to the default value of 30 decimal places.  The numerical precision of each result can be specified individually in its instruction, as illustrated for c, which is computed and displayed to 40 decimals. In this example, the Immediate Window will show

a = 1234567890.0987654321

1 / a = 8.1000000730620006590111459442E-10

e ^ a = 1.60263297808435399238741617564E+536166022

If a had been defined without double quotes in the macro, i.e., as 

    
a = 1234567890.0987654321

then the result for x in the Immediate Window would have read 

a = 1234567890.09877

but those for 1/a and ea would not have been changed. 

For more than occasional use of extended precision, it is often more convenient to use a collective statement, as in  

Sub Test1b()

Dim a, b, c

Dim MP As New Xnumbers

With MP

a = "1234567890.0987654321"

b = .xInv(a)
c = .xExp(a, 40)
Debug.Print "a     = ", a

Debug.Print "1 / a = ", b

Debug.Print "e ^ a = ", c

End With

Set MP = Nothing

End Sub

In each macro, subroutine, or function, the number of decimals used can also be set globally, as with 


DgtMax = 150

Upon completion of the macro, its results can be written onto the spreadsheet, where they will usually be converted back to double precision, and in their display may be limited further by the chosen cell format. Therefore, all calculations that are critical to obtaining high computational accuracy should be performed within the macro, or inside the functions and subroutines it calls, before returning the final result to the spreadsheet. From the user point of view, this is the simplest approach, because in this way the final output doesn’t even show that any special processing was used, although it may generate complaints about the apparent sluggishness of the calculation. 

Below is a macro that takes a number from the spreadsheet, raises it to the tenth power, and returns the answer to the spreadsheet. To use it, place a number in a spreadsheet cell, or highlight any other cell containing a number, and call the macro. The answer will appear in the spreadsheet cell immediately below it, as well as in two message boxes.    

Sub Test2a()

Dim MP As New Xnumbers

Dim a

Dim b As Double
CellValue = Selection.Value

a = MP.xPow(CellValue, 10, 40)

Selection.Offset(1, 0).Select

Selection.Value = a

MsgBox "x ^ 10 = " & a

b = a

Selection.Offset(1, 0).Select

Selection.Value = b

MsgBox "x ^ 10 = " & b

Selection.Offset(-2, 0).Select

Set MP = Nothing

End Sub
The message boxes display both a and b; for an input value of 12.34567, the message boxes will show 

a = 82252032957.44822844133332922210978090071 

and 

b = 82252032957.4482 

respectively, which only differ in that b is dimensioned As Double, which will override the extended precision. Note that the answers shown on the spreadsheet, like that for b in the message box, will have been reduced to double precision, because upon converting the string back to a number, it will be stored in 64 bits. But neither the message box nor the Immediate Window has that constraint, and both can therefore display the extended-precision result in full. 

Still, you may want to find the extended-precision results on the spreadsheet rather than having to copy them from a message box or VBA debugging tool. You can prevent conversion into double precision by concatenating an apostrophe with the numerical value, which will make Excel treat the result as a text string: 

Sub Test2b()

Dim MP As New Xnumbers

Dim a, c

Dim b As Double

CellValue = Selection.Value

a = MP.xPow(CellValue, 10, 40)

Selection.Offset(1, 0).Select

Selection.Value = a

MsgBox "a = " & a

b = a

Selection.Offset(1, 0).Select

Selection.Value = b

MsgBox "a ^ 10 = " & b

c = "'" & a

MsgBox "a ^ 10 = " & c

Selection.Offset(1, 0).Select

Selection.Value = c

Selection.Offset(-3, 0).Select

Set MP = Nothing

End Sub
The column with the receiving cell should be wide enough to accommodate the extended-precision result. Alternatively, you can left-align that receiving cell, and make sure that cells to its right are empty, so that the extended-precision result can overflow into those empty cells. 

Now that you have the tools to write macros using Xnumbers, you will want to know the available Xinstructions. Some of the most useful ones are listed below. Apart from the usual set of elementary instructions, and an ingenious function evaluator, Xnumbers has an extensive set of instructions for manipulating vectors and matrices, and for computations involving complex numbers. In fact, so many of the latter functions are absent from VBA that Xnumbers also contains their standard, double-precision equivalents, distinguishable by the prefix s. 

Function
Instruction
Operation




items within straight brackets are optional




the prefix MP is implied

Elementary single-argument instructions

Negation
.xNeg(a)
–a
Absolute value 
.xAbs(a)
(a(
Inversion
.xInv(a)
1/a
Elementary two-argument operations

Addition
.xAdd(a,b [,Dgt​Max])
a+b
Subtraction
.xSub(a,b [,Dgt​Max])
a–b


or: .xAdd(A1, .xNeg(A2)); do NOT use xAdd(A1,​–A2)

Multiplication
.xMult(a,b [,Dgt​Max])
ab
Division
.xDiv(a,b [,Dgt​Max])
a/b


or: .xMult(A1, .xInv(A2))
Integer division
.xDivint(a,b [,Dgt​Max])
int (a/b)




b ( 0 

Integer modulus
.xMod(a,b [,Dgt​Max])
 b mod (a/b)

Integer power
.xPow(a,n [,Dgt_Max])
a n




if n is a real number, int(n) is used for the exponent 

Arbitrary power
.xExp_Base(a, b [,Dgt​Max])
ab





a > 0 

Square root
.xSqr(a [,Dgt​Max])
(a





a > 0 

Integer root
.xRoot(a,n [,Dgt​Max])
a1/n



if n is non-integer, int(n) is substituted for n     

xroot(100,9,45) ( 1.66810053720005875359979114908865584747919268 

Multi-argument operations

V is a vector, i.e., an array of numbers vi in a contiguous row or column

Summation
.xSum(V [,Dgt​Max])
 vi
Product

.xProd(V [,Dgt​Max])
 vi
Trigonometric functions

all argument are in radians

Sine


.xSin(a [,Dgt​Max])
sin a


Cosine

.xCos(a [,Dgt​Max])
cos a


Tangent

.xTan(a [,Dgt​Max])
tan a


Inverse Sine
.xAsin(a [,Dgt​Max])
arcsin a


Inverse Cosine
.xAcos(a [,Dgt​Max])
arccos a


Inverse Tangent
.xAtan(a [,Dgt​Max])
arctan a


Exponential and logarithmic functions

Exponential
.xExp(a [,Dgt​Max])
ea


Natural Logarithm
.xLn(a [,Dgt​Max])
ln a


Decimal Logarithm
.xLog(a [,Dgt​Max])
log a


Hyperbolic Sine
.xSinh(a [,Dgt​Max])
sinh a




sinh a = ( ex – e–x)/2

Hyperbolic Cosine
.xCosh(a [,Dgt​Max])
cosh a




cosh a = ( ex + e–x)/2

Hyperbolic Tangent
.xTanh(a [,Dgt​Max])
tanh a




tanh a = (ex – e–x)/( ex + e–x)

Inverse Hyperbolic Sine
.xAsinh(a [,Dgt​Max])
arsinh a




arsinh a = ln[a+((a2​+1)]

Inverse Hyperbolic Cosine
.xAcosh(a [,Dgt​Max])
arcosh a




arcosh a = ln[a+((a2​–1)], a > 1

Inverse Hyperbolic Tangent
.xAtanh(a [,Dgt​Max])
artanh a




artanh a = (1/2) ln[(1+a)/(1–a)]

Statistical functions

Factorial
.xFact(n [,Dgt​Max])
n!



for DgtMax < 100, returns result in scientific notation 

Binomial coefficient
.xComb(n, m [,Dgt​Max])

Some useful constants

The brackets are required, even if empty.

DgtMax can go up to 415 for , /2, /4, 2, e, ln(2), ln(10), and 





.xPi( [Dgt​Max])




also: .xPi2( ) for /2, .xPi4( ) for /4, and .x2Pi( ) for 2
e


.xE( [Dgt​Max])
e



also: .xLn2() ( 



and: .xLn10( ) ( 



.xEu( [Dgt​Max])




Euler’s constant, 0.57721566490153286060651209008…

More single-argument instructions

Integer part
.xInt(a)



computes the largest integer smaller than a, e.g., 



xInt(2.14) = 2, xInt(2.99) = 2, xInt(–2.14) = –3

Decimal part
.xDec(a)




computes the decimal part of a, e.g., 



xDec(2.14) = 0.14, xDec(–2.14) = –0.14

Round

.xRound(a, n)




rounds a to n decimal places, e.g., 



xRound(2.14) = 0.14, xDec(–2.14) = –0.14

Comparison
.xComp(a [,b])




.xComp(a, b) = 1 for a > b, = 0 for a = b, = –1 for a < b 



b = 0 when omitted, i.e., .xComp(a) returns the sign of a 

Do NOT include logic operators inside the argument, as then the logic is applied before the extended precision. Therefore, replace If a<b Then by If .xComp(a,b) < 0 Then, 
If a = b Then by If .xComp(a,b) = 0 Then, and

If a>b Then by If .xComp(a,b) > 0 Then.

Formating instructions

Format

.xFormat(a [,Digit_Sep])





formats a in comma-separated groups of Digit_Sep



default: Digit_Sep = 6. For a = 1234567.89012345,



.xFormat(a) = 1,234567.890123,45 and 



.xFormat(a,3) = 1,234,567.890,123,45

Unformat
.xUnformat(a)







Removes formatting commas from a

Matrix and vector operations

We distinguish here between vectors V of dimension m, a positive integer, and matrices M (or A, B, C, etc.) of dimension m(n, where n is another positive integer. A matrix is called square when m = n. Individual matrix elements are denoted by aij.

Absolute value 
.xMatMod(M [,DgtMax])
((M((



Absolute value or modulus 
=

Determinant 
.xMatDet(M [,DgtMax])
(M(



M must be square. Returns “?” when M is singular.

Addition 
.xMatAdd(M1, M2 [,DgtMax])
M1 + M2



M1 and M2 must have the same dimension m(n

Subtraction 
.xMat​Sub(M1, M2 [,DgtMax])
M1 – M2



M1 and M2 must have the same dimension m(n
Scalar product 
.xProdScal(V1, V2 [,DgtMax])
V1 ( V2



V1 and V2 must have the same dimension m
Vector product 
.xProdVect(V1, V2 [,DgtMax])
V1 ( V2



V1 and V2 must have the same dimension m
Multiplication 
.xMat​Mult(M1, M2 [,DgtMax])
M1 M2



when M1 is m(p , M2 must be p(n

Integral power 
.xMat​Pow(M, n [,DgtMax])
Mn  




M must be square, m(m, and n a positive integer
Inversion 
.xMat​Inv(M [,DgtMax])
M–1  




M must be square, m(m

Special matrix functions

LU decomposition 
.xMat​_LU(M [,Pivot] [,DgtMax])
 

M must be square, m(m. Returns the Lower and Upper triangular matrices that satisfy M = L U or, when Pivot is true, M = P L U where P is the permutation matrix. 

LL decomposition 
.xMat​_LL(M [,DgtMax])
 




Cholesky decomposition. M must be square, m(m. 

Linear equation solver .xSysLin(A, B [,DgtMax])





Solves A X = B to yield X = A–1AX = A–1B



A must be square, m(m, and B must be m(1 or m(n 

Similarity transform 
.xMat​BAB(A, B [,DgtMax])





Computes C = B–1 A B. A and B must be square, m(m. 

Gauss-Jordan 
.xGaussJordan(M, n, m, Det, Algo, DgtMax)





performs Gauss-Jordan matrix reduction. 
Special least squares functions

Least squares fit
.xRegLin​_Coeff(y, x [,DgtMax] [,Intercept])


Finds least squares fit of a vector y to a vector or matrix x for both polynomial or multivariate fits, without uncertainty estimates. Optional Intercept forces the fit through the spe-cified y-value Intercept at x = 0. Passes all NIST linear least squares test sets with the default setting DgtMax = 30.

Least squares result
.xRegLin​_Eval(coeff, x [,DgtMax])


Computes the values of ycalc based on the coefficients found (e.g., with .xRegLin_​Coeff) for the specified set of x-values

Functions operating on complex numbers

For functions using complex arithmetic, the prefix MP.Cplx is required. Complex numbers, here denoted by z = a + jb, where j = ((–1), must be defined in terms of their separate,  real and imaginary components, a and b. Note that many of these instructions have the same form as those for real numbers, but are distinguishable by their prefix Cplx. For their double-precision equivalents, in the instruction just change the prefix x to s. 

Negation
.xNeg(z [,DgtMax])
–z



= –(a + jb) = – a – jb 

Absolute value
.xAbs(z [,DgtMax])

(z(



=

Inversion
.xInv(z [,DgtMax])

1/z

Conjugate
.xCon(z [,DgtMax])
z*



= a – jb 

Zero check
.xIsZero(z, tiny [,Dgt​Max])
z ( 0




   (z( =

Addition
.xAdd(z1,z2 [,DgtMax])

z1+z2



= (a1 + a2) + j (b1 + b2) 

Subtraction
.xSub(z1,z2 [,DgtMax])

z1–z2



= (a1 – a2) + j (b1 – b2)

Multiplication
.xMult(z1,z2 [,DgtMax])
z1 z2



= (a1a2–b1b2) + j (a1b2+a2b1) 

Division
.xDiv(z1,z2 [,DgtMax])

z1/z2



=

Integer power
.xPow(z,n [,DgtMax])

zn



=

Integer root
.xRoot(z,n [,k] [,DgtMax])
z1/n



=

Square root
.xSqr(z [,k] [,DgtMax])

z½



=

Exponential
.xExp(z [,DgtMax])

ez



=ea cos(a) + jeb cos(b) 

Natural logarithm
.xLn(z [,DgtMax])

ln z



   = ½ ln(a+jb) + arctan(a/b)

Decimal logarithm
.xLog(z [,DgtMax])

 log z



= ln(z) / ln(10) 

Convert to polar
.xPolar(z [,DgtMax])

 




z = 

Convert to rectangular
.xRect(z [,DgtMax])

 




z = 

Evaluating an algebraic function

There are some practical restrictions with the above list. First, it may lack an operation you may want to perform. Secondly, while the elementary instructions can be nested to evaluate rather complicated algebraic expressions, this approach quickly becomes rather unwieldy. In order to calculate, e.g., the root d = [–b+((b2–4ac)]/2a of the quadratic equation y = ax2 + bx + c one would need to write the instruction root = .xDiv(.xAdd(.xNot(b), .xSqr(.xSub(.xPow(b, 2), .xMult(4, .xMult(a, c))))), .xMult(2,a)), or, broken up, as discr = .xSub(.xPow (b, 2), .xMult(4, .xMult(a, c))), root = .xDiv(.xAdd(.xNot(b), .xSqr(discr)), .xMult(a, 2)), still quite clumsy. Therefore, a convenient function evaluator has been made available, in which f is a function in regular VBA format, and x is either a single number or a vector. 

Function evaluation
xeval(f[,x] [,Dgt​Max] [,Angle])
f (x)

For trigonometric functions: Angle = Rad (default), Deg, or Grad (= Degrees ( 10/9)

x_eval("(1.2345^6.789)") ( 4.17958381726967068826150545493

x_eval("(sqr(5/2)+7^3.21)") ( 517.717800424940440699345256799

x_eval ("(sin(22.5)+3*cos(22.5))") ( 1.66810053720005875359979114908865584747919268

The following example (with the value of a placed inside quotation marks in order to preserve all of its more than fifteen digits) involves a single variable:

Sub Test3a()

Dim MP As New Xnumbers

Dim a, b

With MP

a = "1.234567890123456789"

b = .x_eval("(a ^3.456)", a)
Debug.Print "a = ", a

Debug.Print "a^3.456 = ", b

End With

Set MP = Nothing

End Sub

and yields, in the Immediate Window,

a =  1.234567890123456789

a^3.456 = 2.07145623088457274932940268363

Expressions containing multiple variables must enter the latter in a single matrix, with their names specified as labels in the first matrix column, and their numerical values in the second matrix column, as in 

Sub Test3b()

Dim MP As New Xnumbers

Dim x(1 To 2, 1 To 2) 
'dimension matrix

With MP

x(1, 1) = "a": x(1, 2) = .xE
'specify matrix elements

x(2, 1) = "b": x(2, 2) = .xPi   


b = .x_eval("(a ^ b)", x)

Debug.Print "e =    ", x(1, 2)

Debug.Print "Pi=    ", x(2, 2)

Debug.Print "e^Pi = ", b

End With

Set MP = Nothing

End Sub
 

which produces the following output in the Immediate Window:

e =      2.71828182845904523536028747135

Pi=      3.14159265358979323846264338327

e^Pi = 23.1406926327793214925637779471 

And here, finally, is a macro to compute the root of a quadratic equation with the usual (but, as we will see, sometimes not quite satisfactory) standard expression: 

Sub Test4()

Dim a, b, c, root1, root2

Dim y(1 To 3, 1 To 2)

Dim MP As New Xnumbers

With MP

a = 1

b = 100000

c = 0.000001

y(1, 1) = "a": y(1, 2) = a

y(2, 1) = "b": y(2, 2) = b

y(3, 1) = "c": y(3, 2) = c

root1 = (-b + Sqr(b ^ 2 - 4 * a * c)) / (2 * a)

root2 = .x_eval("((-b + Sqr(b ^ 2 - 4 * a * c)) _

        / (2 * a))", y)
Debug.Print "root1 = ", root1

Debug.Print "root2 = ", root2

End With

Set MP = Nothing

End Sub
with the following results displayed in the Immediate Window:

root1 =  –7.27595761418343E-12

root2 =  –0.00000000001000000000000005

The difference between these two answers shows that, when 4ac « b2, double precision has difficulties evaluating this rather common formula, because it must compute the difference between two nearly equal numbers, b and the square root of b2 – 4ac. In this case, the quadruple precision of Xnumbers is still sufficient to get the answer right. A better way, of course, would be to use an alternative formula that does not involve that difference, such as d = 2c/[–b–((b2–4ac)]. But such an obvious choice might not be available for more complicated problems, in which case extended precision would be a best first defense, before one finds out how to improve the algorithm used. 
Summary

Here are the rules you must follow when writing new macros that use Xnumbers, or when modifying old ones. 

(1) Give the macro a distinct name, such as one starting with an x.

(2) If you modify an already existing macro, keep a copy of the unmodified original. High-precision macros should not be used as routine updates of general-purpose macros, since they are too slow, even when set to the standard double precision of 15 decimals. This is so because much of the extra execution time is up-front, regardless of the precision specified.

(3) Before the dimensions are listed, insert the lines 

Dim MP As New Xnumbers 

With MP

where MP stands for multi-precision. The default (quadruple) precision yields 30 decimals, and need not be specified. If you want a different precision, specify that here instead, as with 

DgtMax = 50 

or, if you so prefer, with an input box that lets you specify the value of DigitsMax from the spreadsheet.  

 (4) Just before the end of the subroutine, insert the lines

End With

Set MP = Nothing

(5) In the dimension statements, remove the dimension specifiers As Single and As Double from any parameter you want to use in extended precision, because these dimension statements will overrule Xnumbers and keep the parameter in their dimension-specified precisions. By leaving their dimensions unspecified (but of course included in the list, otherwise Option Explicit would complain), these parameters will be assumed to be dimensioned As Variant, which is how Xnumbers can use them. 

(6) Then modify all precision-critical instructions to fit the Xnumber format. For complicated expressions, either break up the expression into several simpler parts, or use the xEval function.

By comparing the macros in the xMacroBundle with their parents in the MacroBundle you will see that this need not involve very much work, because you need not rethink the structure of your code, merely some of its instructions. For more specifics, including additional functions and more details, see the Xnumbers_DLL_Ref manual.

*****

In summary, with xnumbers.dll and a few relatively minor modifications of your macros, subroutines, and functions, you can have high-precision computations while the structure of these routines, as well as your data input and output formats, remain unchanged. This can provide a very useful back-up for any custom macros, as it allows you to check the accuracy of your code. It can also help you identify areas of code where it is worthwhile to pay extra attention to the algorithm used. Some of the macros of the MacroBundle have already spawned x-versions which you can find in the xMacroBundle. You are of course welcome to try them out. Please let me know by e-mail, at rdelevie@
bowdoin.edu, if you encounter any problems with them.

